



CIRB  
CIBG

CENTRE D'INFORMATIQUE POUR LA RÉGION BRUXELLOISE  
CENTRUM VOOR INFORMATICA VOOR HET BRUSSELSE GEWEST

# OpenOffice.org 2.0 Calc HOW TO Formules et Fonctions

Réalisation : Filip Lannoye  
Roland Van Wunsel

[flannoye@cirb.irisnet.be](mailto:flannoye@cirb.irisnet.be)  
[rvanwunsel@cirb.irisnet.be](mailto:rvanwunsel@cirb.irisnet.be)

Version 1.0 du 05/05/2006

Texte basé sur OpenOffice.org 2.0.1 EN pour Windows + language packs FR et NL.  
Texte publié sous la licence PDL (<http://www.openoffice.org/licenses/PDL.rtf>).

## Table des matières

1. Formules vs Fonctions
2. Créer des formules
  - 2.1. En référant des cellules de la même feuille
  - 2.2. En référant des cellules de deux feuilles différentes du même classeur
  - 2.3. En référant des cellules de deux classeurs différents
  - 2.4. Opérateurs
  - 2.5. Editer une formule
3. Adresses relatives, absolues et mixtes
  - 3.1. C'est quoi le problème ?!
  - 3.2. La solution magique du signe "\$"
4. Utiliser des fonctions
  - 4.1. Comment utiliser les fonctions ?
  - 4.2. Quelques fonctions statistiques
  - 4.3. Quelques fonctions mathématiques
  - 4.4. Quelques fonctions logiques
  - 4.5. Quelques fonctions de date et heure
  - 4.6. Quelques fonctions de recherche
  - 4.7. Exemples intégrés
5. Crédits
6. Licence

Avenue des Arts 21 Kunstlaan  
Bruxelles 1000 Brussel  
T 32 2 282 47 70 F 32 2 230 31 07  
[www.cirb.irisnet.be](http://www.cirb.irisnet.be) [www.cibg.irisnet.be](http://www.cibg.irisnet.be)

## 1. Formules vs Fonctions

---

Tant les formules que les fonctions servent à faire des calculs dans un tableur. Les deux commencent obligatoirement par le signe "=" pour référencer ensuite les cellules qui contiennent les valeurs à prendre en compte dans le calcul. De préférence, on évite au maximum les références fixes dans les formules (chiffres absolus) pour **référencer les adresses de cellule**. L'avantage évident de cela est que les valeurs dans une cellule peuvent changer et que les formules qui y font référence sont calculées correctement, alors que si l'on met un chiffre fixe dans une formule, il faut le modifier à la main pour que la formule change. Le risque d'erreurs serait très grand.

"Formule" = Calcul mathématique combinant des cellules par un opérateur mathématique.

Syntaxe : =<Cellule><opérateur><Cellule>  
p.ex. =(A1+A2+A3)/3

"Fonction" = Calcul mathématique prédéfini et stocké sous un certain nom.

Exemple : moyenne = somme d'une série de valeurs divisée par le nombre de valeurs de la série

Syntaxe : =<Nom de la fonction>(Arguments)  
p.ex. =MOYENNE(A1:A3)

[Retour](#)

## 2. Créer des formules

---

### 2.1. En référant des cellules de la même feuille

Intérêt : faire des petites formules rapides qui font référence à des cellules qui sont typiquement réparties sur la feuille. Par contre, pour des calculs faisant appel à des cellules contiguës, on utilise de préférence des fonctions (voir plus bas).

Procédure :

- Cliquez sur la cellule dans laquelle vous voulez écrire la formule.
- Tapez le signe égal "=".
- Cliquez sur la première cellule contenant une valeur.
- Tapez l'opérateur mathématique nécessaire.
- Cliquez sur la deuxième cellule contenant une valeur.
- Si c'est la dernière cellule à référencer, validez la formule par "**Enter**". Sinon, remettez un opérateur et recliquez sur la cellule qu'il faut. Etc.

Exemple :

- A1 contient le chiffre 12 ;
- C3 contient le chiffre 21 ;
- D10 contient le chiffre 15% ;

- Dans E25, vous devez additionner les valeurs des cellules A1 et C3 pour en soustraire une réduction que vous calculez à l'aide du taux en D10.
- Formule en E25 : **=(A1+C3)\*(1-D10)**
- Attention aux calculs faisant appel à des *pourcentages* : typiquement, on fait des calculs du genre "1+x%" ou "1-x%", un pourcentage n'étant rien d'autre qu'un chiffre exprimé en centièmes (donc 15% = 0,15) et formaté en tant que pourcentage, en même temps que "1" est égal à 100%. La formule réelle est donc bien "100%+15%" ou "100%-15%", ce que d'habitude, on exprime comme "\*1,15" ou "\*0,85". L'avantage absolu ici consiste à pouvoir changer le taux en D10 de façon dynamique.

[Retour](#)

## 2.2. En référant des cellules de deux feuilles différentes du même classeur

Exemple :

- A1 de Feuille1 contient la valeur 30 ;
- B2 de Feuille2 contient la valeur 50 ;
- Dans E1 de Feuille1, on aurait besoin de multiplier ces deux chiffres.

Procédure :

- Cliquez sur la cellule E1 de Feuille1 qui est la cellule dans laquelle vous voulez écrire la formule.
- Tapez le signe égal "=".
- Cliquez sur la cellule A1 de la même feuille.
- Tapez l'opérateur mathématique "\*".
- En bas de l'écran, cliquez sur l'onglet "Feuille2" pour changer de feuille.
- Cliquez sur la cellule B2.
- Validez la formule par "**Enter**". Vous serez de retour à la Feuille1, où vous verrez le résultat du calcul dans la cellule E1.
- Syntaxe : **=A1\*Feuille2.B2**
- Attention à la *référence à une feuille* dans une formule : elle est séparée de l'adresse de cellule par le signe "." (**point**).

[Retour](#)

## 2.3. En référant des cellules de deux classeurs différents

Exemple :

- A1 de Feuille1 de Classeur1 contient la valeur 20 ;
- B2 de Feuille2 de Classeur 2 contient la valeur 350 ;
- Dans C1 de Feuille1 de Classeur1, on aurait besoin de multiplier ces deux chiffres.
- Classeur1 et Classeur2 sont sauvegardés dans le dossier D:\Data\Exemples.

Procédure :

- Les deux fichiers doivent être ouverts.
- Activez le fichier "Classeur2" en cliquant sur le bouton correspondant de la "Barre des Tâches" de Windows.
- Cliquez sur la cellule B2 de Feuille2 qui contient la valeur à référencer.
- Copiez cette cellule.

- Activez le fichier "Classeur1" en cliquant sur le bouton correspondant de la "Barre des Tâches" de Windows.
- Cliquez sur la cellule C1 de Feuille1 qui est la cellule dans laquelle vous voulez écrire la formule.
- Cliquez sur le menu "**Edition – Collage spécial** – cochez l'option **Lier**" et validez par "**OK**".
- Le curseur se trouve toujours sur la cellule C1. Vous lisez alors la formule suivante sur la barre de formule :  
**=DDE("soffice";"D:\Data\Exemples\Classeur2.ods";"Feuille2.B2")**
- Cliquez à la fin de cette référence sur la barre de formule.
- Tapez l'opérateur mathématique "\*".
- Cliquez sur la cellule A1.
- Validez la formule par "**Enter**".
- Syntaxe :  
**=DDE("soffice";"D:\Data\Exemples\Classeur2.ods";"Feuille2.B2")\*A1**
- Attention à la *référence à un autre fichier* dans une formule : en réalité, il faut copier un **lien DDE** qui pointe vers la cellule source et construire ensuite la formule autour de cette référence.

[Retour](#)

## 2.4. Opérateurs

### A. Liste des opérateurs mathématiques

- + Additionner
- Soustraire
- \* Multiplier
- / Diviser
- () Changer la priorité de calcul
- ^ Exposant
- Attention aux règles de priorité en mathématiques : \* et / vont avant + et -, donc si vous voulez multiplier la somme de trois chiffres, il faut d'abord faire l'addition de ces chiffres entre parenthèses pour réaliser d'abord la somme et ensuite seulement faire la multiplication.
- Exemple :  $2 * (3 + 4^2) - 1 = 37$

### B. Liste des opérateurs de comparaison

- = X est égal à Y
- > X est plus grand que Y
- >= X est plus grand ou égal à Y
- < X est inférieur à Y
- <= X est inférieur ou égal à Y
- <> X est différent de Y

[Retour](#)

## 2.5. Editer une formule

- Pour éditer une formule, cliquez sur la cellule qui contient la formule et poussez sur la touche "**F2**". L'intérêt est que vous éditez la formule directement dans la cellule et que vous voyez clairement toutes les références d'adresses de cellule grâce à un système de couleurs, où chaque référence isolée est marquée dans une certaine couleur et que la plage de cellules correspondante est entourée dans la même couleur. Cela rend la vérification et l'édition de formule particulièrement facile.
- Si la formule est un peu longue ou que la cellule contenant la formule est entourée de beaucoup d'information, il peut être plus difficile d'interpréter la formule directement dans la cellule. C'est pourquoi vous pouvez aussi éditer la formule en cliquant dedans sur la barre de formule.

[Retour](#)

## 3. Adresses relatives, absolues et mixtes

---

### 3.1. C'est quoi le problème ?!

Un tableur est organisé en **colonnes** et **lignes**. Chaque croisement entre une colonne et une ligne fait une cellule. Dans l'esprit d'un tableur, on crée des tableaux du haut à gauche vers le bas à droite. En même temps, quand on crée une formule, on crée une formule pour la première ligne ou la première colonne et on la **copie** vers les autres lignes ou les autres colonnes. Ainsi, on obtient le même type de formule partout. Par la suite, tout changement de formule se ferait dans la première cellule de la série pour la recopier de nouveau ensuite. Cela réduit beaucoup le risque d'erreurs.

A force de copier une formule **de gauche à droite**, l'adresse de la cellule hébergeant la formule change aussi quant à sa **référence de colonne**. Ainsi, une cellule A3 qui contient la formule " $=A1+A2$ " et qu'on copie à droite, devient B3, puis C3 et ainsi de suite. A force de copier donc la formule sur l'axe horizontal, la référence de colonne incrémente de 1, alors que la référence de ligne ne change pas (3 dans tous les cas de l'exemple). Ceci est appelé "**copier de façon relative**". Quand vous vérifiez les formules contenues dans les cellules B3 et C3, vous verrez respectivement " $=B1+B2$ " et " $=C1+C2$ ".

Si vous copiez une formule dans D1 vers le bas, vous la copiez vers les cellules D2, D3 et suivantes. A force de copier une formule **de haut en bas**, l'adresse de la cellule hébergeant la formule change donc quant à sa **référence de ligne**. Si la formule en D1 est " $=A1+B1+C1$ ", alors dans les cellules D2 et D3 vous lisez maintenant respectivement " $=A2+B2+C2$ " et " $=A3+B3+C3$ ". Il s'agit de nouveau d'une copie relative, où les références de cellules changent d'après la direction de la copie.

Par contre, quand une partie de la formule fait référence à une valeur absolue, cette référence ne peut pas changer quand on copie la formule. Ainsi, si vous mettez un taux de TVA de 21% dans la cellule F1 et que vous voulez augmenter vos totaux de colonne (cellules A3:C3) de TVA, la formule en A3 devient " $=(A1+A2)*(1+F1)$ ". Si vous copiez ensuite cette formule à droite, vous obtenez " $=(B1+B2)*(1+G1)$ " et " $=(C1+C2)*(1+H1)$ ". A force de copier la formule de façon relative, toutes les références de colonne augmentent de 1, y inclus la référence à la TVA en F1, et cela donne un résultat erroné, car on devrait avoir F1 partout !

Il en va de même pour la copie verticale : si vous augmentez vos totaux de ligne (cellules D1:D3) de 21% de TVA dans F1, vous obtenez les formules " $=(A1+B1+C1)*(1+F1)$ ",

"=(A2+B2+C2)\*(1+F2)" et "=(A3+B3+C3)\*(1+F3)", ce qui est de nouveau erroné, car le F1 devait rester F1 partout !

[Retour](#)

### 3.2. La solution magique du signe "\$"

Pour éviter qu'une référence change quand vous copiez une formule à droite ou vers le bas, vous pouvez insérer le signe dollar "\$" devant la référence de cellule à bloquer. A force d'insérer ce signe, vous rendez une adresse de cellule **absolue** et donc invariable.

Attention : le signe "\$" est à insérer devant la partie de l'adresse de cellule qui ne peut pas changer à la copie. Tout dépend donc du sens de la copie pour savoir si vous mettez ce signe devant la partie "colonne" ou "ligne" ou les deux en même temps.

Ainsi, pour reprendre les deux exemples du point précédent, dans la formule de la cellule A3, le "F" de "F1" changeait en "G" et "H", mais le "1" ne changeait pas. Autrement dit, à force de copier une formule à droite, la référence "colonne" d'une adresse de cellule change, mais pas la ligne. Si la référence "colonne" ne peut pas changer, il faut mettre un dollar devant cette référence et recopier de nouveau la formule. Vous obtiendrez :

A3	=(A1+A2)*(1+\$F1)
B3	=(B1+B2)*(1+\$F1)
C3	=(C1+C2)*(1+\$F1)

De la même façon, dans la formule de la cellule D1, il faut mettre un signe "\$" devant le "1" de "F1" pour bloquer la référence de ligne. Quand vous copiez ensuite de nouveau cette formule, vous obtiendrez :

D1	=(A1+B1+C1)*(1+F\$1)
D2	=(A2+B2+C2)*(1+F\$1)
D3	=(A3+B3+C3)*(1+F\$1)

Pour insérer le signe "\$", vous cliquez au bon endroit dans la formule sur la barre de formule et soit vous le tapez, soit vous poussez sur la combinaison de touches "**SHIFT+F4**". Si vous enfoncez ces touches plusieurs fois de suite, différentes combinaisons apparaissent (en cycle) : **\$F\$1, F\$1, \$F1 et F1**.

Il est vrai que dans les deux cas, vous auriez pu mettre un signe "\$" devant les deux références pour obtenir **\$F\$1**, mais dans le cas de la copie horizontale, le "\$" devant "1" n'apporte rien et dans le cas de la copie verticale, le "\$" devant "F" n'apporte rien, car ces références ne changent pas de toute façon.

Strictement parlant, les deux signes "\$" sont uniquement nécessaires si vous faites référence à une cellule dans une formule qui sera copiée aussi bien à droite qu'en bas. En général, c'est une bonne pratique de ne mettre que le signe qu'il faut. Ainsi, vous contrôlez parfaitement les copies de formules.

En résumé :

<b>Adresse de cellule</b>	<b>Type d'adresse</b>	<b>Ce qui est bloqué</b>
<b>F1</b>	Adresse relative	Rien
<b>F\$1</b>	Adresse mixte	Ligne (pour la copie verticale)
<b>\$F1</b>	Adresse mixte	Colonne (pour la copie horizontale)
<b>\$F\$1</b>	Adresse absolue	Ligne et Colonne (pour la copie dans les deux sens)

*A retenir* : **mettez le signe "\$" devant la référence qui ne peut pas changer**, c'est-à-dire devant la ligne pour la copie vers le bas, devant la colonne pour la copie à droite et devant les deux pour une copie vers le bas et à droite.

La question des adresses relatives et absolues est omniprésente. Partout où il est question d'adresses de cellule, il faut vous poser la question s'il faut adapter les références ou pas. Ainsi, vous en aurez besoin dans des formules et des fonctions, mais aussi dans les noms de plage, la validité de données, le formatage conditionnel et les filtres.

[Retour](#)

## 4. Utiliser des fonctions

### 4.1. Comment utiliser les fonctions ?

#### A. Le grand classique : la fonction "Somme"


Pour additionner des valeurs de cellules contiguës, plutôt que d'utiliser une formule, vous avez intérêt à utiliser la fonction "Somme" et de la copier dans les bonnes colonnes et lignes. Comme la somme est la fonction la plus couramment utilisée, il existe un bouton pour l'insérer rapidement.

Procédure :

- Mettez le curseur dans la cellule tout de suite en dessous de la série de cellules à additionner pour faire la somme verticale ou tout de suite à droite pour la somme horizontale.
- Cliquez sur le bouton "**Somme**"  $\Sigma$  de la barre de formule. La fonction est inscrite avec entre parenthèses la série de cellules sélectionnée automatiquement.
- Si la série suggérée est correcte, validez par "**Enter**" ou par un clic sur le bouton "**Appliquer**"  de la barre de formule.
- Si la série suggérée n'est pas correcte, corrigez sa référence en manipulant la sélection en surbrillance. Ainsi, à partir du petit plus au coin droit inférieur, il y a un moyen de redimensionner la sélection. Quand vous pointez sur la sélection entourée en bleu, le pointeur de la souris se transforme en une petite main. Si vous cliquez et glissez maintenant, vous pouvez déplacer la sélection de base.

#### B. Chercher une fonction à insérer

- Mettez le curseur sur la cellule dans laquelle vous voulez insérer la fonction.
- Cliquez sur le bouton "**Assistant Fonctions**"  de la barre de formule.




- Sous "**Catégorie**", sélectionnez la catégorie de fonctions qu'il vous faut. Cela filtre la liste des fonctions qui apparaît sous "**Fonction**".
- Sous "**Fonction**", sélectionnez la fonction désirée. A droite, vous voyez apparaître la syntaxe de la fonction, ainsi qu'une brève description.
- Cliquez sur le bouton "**>>Suivant**" pour aller dans l'étape de la configuration de la fonction.
- Saisissez les arguments nécessaires en vous servant au maximum de la souris. S'il est question d'adresses de cellule, préférez les sélectionner plutôt que de taper leurs adresses pour éviter toute erreur.
- Exemple de la fonction "Moyenne" :

	A	B	C	D	E	F	G	H
1	100							
2	200							
3	300							
4	400							
5								

- Validez l'ensemble par "OK".
- *Attention* : les **arguments obligatoires** sont marqués en caractères **gras** ; les optionnels en caractères normaux.

### C. Insérer une fonction dernièrement utilisée

- Mettez le curseur sur la cellule dans laquelle vous voulez insérer la fonction.
- Sur la barre de formule, cliquez sur le signe égal "=".
- Revenez à gauche sur la barre de formule et ouvrez la **liste déroulante**. Sélectionnez la fonction désirée. Elle s'inscrit tout de suite dans la cellule.



- Renseignez les paramètres requis et validez.
- Si vous voulez utiliser la fenêtre de l'assistant pour compléter la fonction, cliquez sur le bouton "**Assistant Fonctions**" , renseignez les paramètres nécessaires et validez la fonction.

### D. Editer une fonction insérée

- Pour éditer une fonction déjà insérée, cliquez sur la cellule contenant cette fonction.
- Recliquez sur le bouton "**Assistant Fonctions**" , modifiez les arguments nécessaires et validez par "OK".

### E. Utiliser l'aide

- Si vous n'êtes pas sûr du choix de la fonction ou que vous voulez voir un exemple de la fonction choisie, à partir de la deuxième étape de l'Assistant Fonctions, cliquez sur le bouton "**Aide**". La page d'aide concernant la fonction s'ouvre.

### F. Quelles fonctions existe-t-il ?

- Dans la première étape de l'Assistant Fonctions, vous pouvez parcourir les fonctions par catégorie pour savoir ce qui existe. La plupart des noms de fonctions sont assez intelligibles.  
*Astuce* : quand vous cliquez sur une fonction de la liste et que vous tapez la première lettre d'une fonction cherchée, la liste défile rapidement.
- Une autre approche consiste à creuser dans les pages d'aide. Poussez pour cela sur la touche "**F1**" ou cliquez sur le menu "**Aide – Aide de OpenOffice.org**".
- A partir de la première page d'aide, suivez le lien "**Fonctions de OpenOffice.org Calc**". Vous verrez apparaître la **liste des catégories de fonctions** pour ensuite descendre dans chacune des catégories.
- Toujours à partir de l'aide, vous pouvez également chercher le nom d'une fonction connue directement. Si jamais la fonction ne vous convient pas, vous pouvez suivre

les liens en dessous de la fonction pour remonter dans la catégorie des fonctions voisines.

- Si vous voulez consulter la liste complète et détaillée de toutes les fonctions existantes en Calc, téléchargez le guide "**Fonctions de Calc**" disponible dans les Guides du site <http://fr.openoffice.org> à l'adresse <http://fr.openoffice.org/Documentation/Guides/Indexguide.html>.

[Retour](#)

## 4.2. Quelques fonctions statistiques

	A	B	C
1		100	
2		200	
3			
4		ABC	
5		Catégorie 2	
6		300	
7			
8	Nombre de valeurs numériques	3	≙NB(B1:B6)
9	Nombre de cellules remplies	5	≙NBVAL(B1:B6)
10	Chiffre le plus bas de la série	100	≙MIN(B1:B6)
11	Chiffre le plus haut de la série	300	≙MAX(B1:B6)
12	Moyenne des chiffres	200	≙MOYENNE(B1:B6)

[Retour](#)

## 4.3. Quelques fonctions mathématiques

	A	B	C
1		100	
2		-200	
3		300	
4		-400	
5		500	
6	Nombre de chiffres négatifs	2	≙NB.SI(B1:B5;"<0")
7	Nombre de chiffres positifs	3	≙NB.SI(B1:B5;">=0")
8	Somme des chiffres négatifs	-600	≙SOMME.SI(B1:B5;"<0")
9	Somme des chiffres positifs	900	≙SOMME.SI(B1:B5;">=0")

- *Attention* : le critère des fonctions NB.SI et SOMME.SI s'écrit entre guillemets !

	A	B	C
1		100	€
2		200	\$
3		300	€
4		400	\$
5		500	€
6	Somme des chiffres en €	900	≙SOMME.SI(C1:C5;"€";B1:B5)
7	Somme des chiffres en \$	600	≙SOMME.SI(C1:C5;"\$";B1:B5)

- Dans ce deuxième exemple, la fonction "SOMME.SI" prend trois arguments : d'abord, on repère les cellules dans la colonne C qui contiennent le symbole monétaire "€", puis on fait la somme des chiffres correspondants de la colonne B.

[Retour](#)

#### 4.4. Quelques fonctions logiques

	A	B	C	D	E
1		100			
2		200			
3					
4		ABC			
5		Catégorie 2			
6		300			
7					
8	Nombre de valeurs numériques	3	=NB(B1:B6)		
9	Nombre de cellules remplies	5	=NBVAL(B1:B6)		
10	Chiffre le plus bas de la série	100	=MIN(B1:B6)		
11	Chiffre le plus haut de la série	300	=MAX(B1:B6)		
12	Moyenne des chiffres	200	=MOYENNE(B1:B6)		
13	Contient si Moyenne>250	Pas content	=SI(B12>250;"Content";"Pas content")		
14	Contient si Minimum>=100 et Maximum >=300	Content	=SI(ET(B10>=100;B11>=300);"Content";"Pas content")		

- La fonction "**SI**" prend trois arguments : le cas à tester, la valeur à rendre si le test est positif et la valeur à rendre si le test est négatif.
- Dans B14, la fonction logique "ET" a été ajoutée pour combiner deux critères qui doivent être respectés tous les deux pour que le test soit valide. Ainsi, si le minimum de la série (B10) donne au moins 100 et le maximum (B11) au moins 300, on affiche la valeur "Content", sinon "Pas content".
- "**ET**" sert à combiner des critères qui doivent tous être respectés pour que le résultat du test donne "VRAI".  
"**OU**" sert à combiner des critères dont il suffit qu'il y en ait un qui soit respecté pour que le résultat du test donne "VRAI".

[Retour](#)

#### 4.5. Quelques fonctions de date et heure

	A	B	C
1	Date de référence : Aujourd'hui	03/05/2006	=AUJOURDHUI()
2	Année de la date de référence	2006	=ANNEE(B1)
3	Mois de la date de référence	5	=MOIS(B1)
4	Jour de la date de référence	3	=JOUR(B1)
5	Jour de la semaine de la date de référence	3	=JOURSEM(B1;2)
6	Numéro de la semaine de la date de référence	18	=NO.SEMAINE(B1;2)
7	Dernier jour du mois de la date de référence (typique pour le calcul d'une date d'échéance)	31/05/2006	=FIN.MOIS(B1;0)

- Attention au formatage des dates, car le programme les traite comme des chiffres !  
Pour changer le formatage, allez dans "**Format – Cellules – onglet Nombre –**

**catégorie Date**" et choisissez le format qui convient, typiquement "DD/MM/AAAA", c'est-à-dire deux positions pour le jour, deux pour le mois et quatre pour l'année.

[Retour](#)

#### 4.6. Quelques fonctions de recherche


	A	B	C	D	E	F	G	H	I
1	N° Produit	Description	Prix unitaire	Quantité	Total				
2	100	PC	700,00 €	20	14.000,00 €				
3	200	Ecran	250,00 €	15	3.750,00 €				
4	300	Clavier	25,00 €	20	500,00 €				
5	200	Ecran	250,00 €	30	7.500,00 €				
6	500	Imprimante	150,00 €	10	1.500,00 €				
7									
8		=RECHERCHEV(\$A2;\$G\$12:\$I\$16;2)							
9									
10			=RECHERCHEV(\$A2;\$G\$12:\$I\$16;3)						
11							N° Produit	Description	Prix unitaire
12							100	PC	700,00 €
13							200	Ecran	250,00 €
14							300	Clavier	25,00 €
15							400	Souris	10,00 €
16							500	Imprimante	150,00 €

- La fonction **"RECHERCHEV"** est la fonction de recherche par excellence. Elle situe l'élément donné, 100 dans A2 de l'exemple, dans la première colonne de la matrice (\$G\$12:\$I\$16) (première position dans ce cas) pour rendre la valeur correspondante de la colonne demandée (2, car on veut rendre la description dans la colonne B ; elle se trouve dans la deuxième colonne de la matrice).
- L'utilisation des signes **"\$"** s'avère judicieux dans ce contexte : la matrice est ainsi rendue fixe dans les formules copiées. Quant à "\$A2", plutôt que "A2", c'est une astuce dans ce cas-ci, car ainsi, on peut facilement copier la formule initiale de la cellule B2 à C2 et changer la colonne de référence 2 en 3 pour afficher le prix unitaire au lieu de la description. Le reste de la formule reste identique et c'est donc plus rapide que de réencoder toute la fonction par l'assistant.
- Une alternative très souvent utilisée dans ce contexte consisterait à nommer la matrice et d'utiliser ce nom à la place des adresses de cellules dans la fonction. Ainsi, pour la cellule B2, si on nomme la matrice \$G\$12:\$I\$16 "catalogue", la formule deviendrait **"=RECHERCHEV(\$A2;catalogue;2)"**, ce qui est nettement plus facile à lire. Pour créer un **nom de plage**, sélectionnez la matrice et cliquez dans la première zone de la barre de formule ("**Zone de nom**"), où l'adresse de la sélection actuelle apparaît. Tapez le nom "catalogue" et validez par "Enter". Pour gérer le nom après coup, utilisez le menu **"Insérer – Noms – Définir"**.
- Si la valeur donnée à repérer dans la liste n'y apparaît pas, la fonction "RECHERCHEV" la situe entre les éléments existants. Pour que cette recherche par tranche fonctionne correctement, il faut que la première colonne de la matrice soit triée par ordre ascendant. Ainsi, une valeur donnée 475 à situer dans une liste qui incrémente par centaine, rendrait la valeur correspondante de 400, car le seuil de 500 n'est pas atteint. Par contre, il existe des contextes, dont le petit bon de commande de l'exemple, où il n'est pas permis de rendre une valeur pour un code inexistant. Dans ce cas-ci, si on tape un code inexistant, on devrait afficher un message d'erreur pour attirer l'attention

de l'utilisateur sur cette erreur. A ce moment-là, soit il met un bon code, soit il étend la liste source par le nouveau code. Pour que la fonction RECHERCHEV tienne compte de cette contrainte, il faut ajouter un quatrième argument "0":

	A	B	C	D	E	F	G	H	I
1	N° Produit	Description	Prix unitaire	Quantité	Total				
2	250	#NA	#NA	20	#NA				
3	200	Ecran	250	15	3.750,00 €				
4	300	Clavier	25	20	500,00 €				
5	200	Ecran	250	30	7.500,00 €				
6	500	Imprimante	150	10	1.500,00 €				
7									
8		=RECHERCHEV(\$A2;catalogue;2;0)							
9									
10			=RECHERCHEV(\$A2;catalogue;3;0)						
11									
12							N° Produit	Description	Prix unitaire
13							100	PC	700,00 €
14							200	Ecran	250,00 €
15							300	Clavier	25,00 €
16							400	Souris	10,00 €
							500	Imprimante	150,00 €

- Le code 250 n'étant pas un code valide, on n'affiche pas la description et le prix du code 200, mais l'erreur "#NA" ("Not available", pas disponible) grâce au dernier argument "0".
- Une variante de la fonction RECHERCHEV est "**RECHERCHEH**". Elle fonctionne de la même façon, sauf que la liste de référence est organisée en lignes plutôt qu'en colonnes et que la première ligne devient donc la ligne de recherche.
- Un point très fort d'OpenOffice.org consiste en l'édition d'une fonction encodée. Si vous cliquez sur la cellule B2 et que vous cliquez sur le bouton "Assistant Fonctions" de la barre de formule pour l'éditer, vous arrivez dans le deuxième onglet de la fenêtre, qui affiche la structure arborescente de la formule complète. Quand vous cliquez sur l'élément de votre choix dans cette structure, le curseur se positionne tout de suite au bon endroit pour faciliter l'édition. Cette technique est particulièrement intéressante pour l'édition de fonctions intégrées (voir point suivant).



[Retour](#)


## 4.7. Exemples intégrés

- Le grand défi de l'utilisation des fonctions consiste à **combiner plusieurs fonctions** pour répondre à une question. Très souvent, la combinaison intelligente des fonctions présentées dans ce document, vous permet de traiter la plupart des questions. Le réel problème n'est pas tellement le choix de la bonne fonction, ni l'encodage de ces arguments, mais la façon de **décortiquer votre question complexe** en questions simples auxquelles vous pouvez répondre par des fonctions de base en Calc.
- Vous utiliserez très fréquemment la fonction logique "SI" pour tester la validité d'un cas concret. En termes d'approche, il vaut très souvent mieux commencer par la réelle question et l'enrober ensuite de la fonction "SI" pour conditionner le fait de calculer ou pas.
- L'exemple suivant est présenté en trois variantes :
  - Tableau 1 : le tableau de base reprend des chiffres de bases qui sont additionnés par pays et par produit. Les totaux des pays sont exprimés en pourcentage de répartition par rapport au total général.
  - Tableau 2 : si le tableau est vidé (pour en faire un modèle de tableau par exemple), les pourcentages posent problème, car le total général étant 0, on fait des divisions par 0, ce qui n'est pas possible. L'idée serait de conditionner le fait de calculer le pourcentage ou pas.
  - Tableau 3 : parmi les solutions possibles, on conditionne d'abord le fait de calculer la somme par pays en forçant ainsi l'encodage correct de tous les éléments avant d'exécuter les calculs (= moyen de contrôler la validité des données). Puis, le calcul du pourcentage dépend du total général. Dès qu'une colonne complète est remplie, il y aura une somme et donc aussi un total général supérieur à 0. Là seulement, le problème de la division par 0 disparaît. Le test du calcul du pourcentage consiste donc simplement à voir si le total général n'est pas 0.

	A	B	C	D	E
1		Belgique	France	Allemagne	Total
2	PC	1500	2000	3000	6500
3	Ecrans	1000	5000	4000	10000
4	Imprimantes	3000	4000	1500	8500
5	Total	5500	11000	8500	25000
6	%/Total Pays	22,00%	44,00%	34,00%	100,00%
7					
8					
9		Belgique	France	Allemagne	Total
10	PC				0
11	Ecrans				0
12	Imprimantes				0
13	Total	0	0	0	0
14	%/Total Pays	#VALEUR !	#VALEUR !	#VALEUR !	#VALEUR !
15					
16					
17		Belgique	France	Allemagne	Total
18	PC	100			100
19	Ecrans	200			200
20	Imprimantes				0
21	Total				0
22	%/Total Pays	0,00%	0,00%	0,00%	0,00%

	A	B	C	D	E
1		Belgique	France	Allemagne	Total
2	PC	1500	2000	3000	=SOMME(B2:D2)
3	Ecrans	1000	5000	4000	=SOMME(B3:D3)
4	Imprimantes	3000	4000	1500	=SOMME(B4:D4)
5	Total	=SOMME(B2:B4)	###	###	=SOMME(B5:D5)
6	%/Total Pays	=B5/\$E5	=C5/\$E5	=D5/\$E5	=SOMME(B6:D6)
7					
8					
9		Belgique	France	Allemagne	Total
10	PC				=SOMME(B10:D10)
11	Ecrans				=SOMME(B11:D11)
12	Imprimantes				=SOMME(B12:D12)
13	Total	=SOMME(B10:B12)	###	###	=SOMME(B13:D13)
14	%/Total Pays	=B13/\$E13	=C13/\$E13	=D13/\$E13	=SOMME(B14:D14)
15					
16					
17		Belgique	France	Allemagne	Total
18	PC	100			=SOMME(B18:D18)
19	Ecrans	200			=SOMME(B19:D19)
20	Imprimantes				=SOMME(B20:D20)
21	Total	=SI(NB(B18:B20)=3;SOMME(B18:B20);"")	=SI(NB(C18:))	=SI(NB(D18:))	=SOMME(B21:D21)
22	%/Total Pays	=SI(\$E21<>0;B21/\$E21;0)	###	###	=SOMME(B22:D22)


- La formule en B21 se lit comme suit : si toutes les données de la Belgique sont remplies, on fait la somme, sinon on affiche un blanc (deux guillemets). Pour voir si toutes les données sont remplies, on ne va pas tester chacune des cellules, mais on utilise la fonction "NB", qui compte le nombre de cellules remplies par un chiffre dans une série de données. La partie "NB(B18:B20)" donne un chiffre entre 0 et 3, qu'on teste par la suite pour voir s'il est égal à 3. La partie "test" de la fonction SI teste donc si 0, 1, 2 ou 3 est égal à 3 et ce n'est que dans le dernier cas que le test donne "VRAI" est que la partie "SOMME(B18:B20)" est calculée. Dans le cas contraire, la partie fausse est exécutée et rend les deux guillemets, soit un blanc. Quand vous éditez la fonction imbriquée par l'Assistant Fonctions, vous obtenez la structure suivante :



- Attention** : pour **imbriquer une fonction** à partir de l'assistant, il faut cliquer sur le bouton "fx" à côté de la zone nécessaire et recommencer la sélection de fonction et le paramétrage. Pour revenir ensuite au niveau supérieur, éviter de cliquer sur "OK", mais

cliquez au bon endroit de la syntaxe de la fonction en construction dans la zone "Formule" en bas de la fenêtre.

- La formule en B22 est même plus simple : elle vérifie si le total en E21 (avec "\$", car on copie la formule à droite !) est différent de 0. Si c'est le cas, c'est qu'au moins un pays était rempli correctement et alors là, cela a du sens de calculer un pourcentage. Tant que le total général reste à 0, cela n'a pas de sens et on affiche donc 0.  
Vue structurée :



- Astuces fréquemment appliquées :

<b>Comment fait-on pour ...</b>	<b>Fonction à utiliser</b>
Voir si toute une série de cellules est remplie	<b>NB</b> ; cela donne le nombre absolu de cellules remplies par un chiffre ; tester ce chiffre par une fonction SI p.ex. SI(NB(A1:A3)=3;"OK";"NOK")
Voir si toutes les valeurs d'une série sont supérieures à 100	100 est le chiffre minimal autorisé. Extraire le chiffre le plus bas de la série par la fonction <b>MIN</b> et tester ce chiffre par rapport à 100. D'office, les autres chiffres de la série sont aussi au moins 100. p.ex. SI(MIN(A1:A3)>=100;"OK";"NOK")
Voir si au moins un chiffre de la série dépasse 1000	Tester le chiffre le plus grand de la série par la fonction <b>MAX</b> pour voir s'il dépasse le seuil. p.ex. SI(MAX(A1:A3)>=1000;"OK";"NOK")
Voir si un élément cherché existe ou pas et afficher un message en fonction du résultat	Utiliser la fonction d'information <b>ESTNA</b> ("est #NA"), autrement dit, exécuter la fonction pour voir si elle rend l'erreur #NA. p.ex. SI(ESTNA(RECHERCHEV(\$A2;catalogue;2;0); "N'existe pas!";RECHERCHEV(\$A2;catalogue;2;0))

[Retour](#)

## 5. Crédits

---

Auteur : **Filip Lannoye**

Remerciement : **Roland van Wunsel, CIRB**

Intégré par : **Tony Galmiche sur le site <http://fr.openoffice.org>**

Contacts : **Centre d'Informatique pour la Région Bruxelloise – [www.cirb.irisnet.be](http://www.cirb.irisnet.be)**

Traduction :

Historique des modifications :

<b>Version</b>	<b>Date</b>	<b>Commentaire</b>
1.0	05/05/2006	Première version pour publication

[Retour](#)

## 6. Licence

---

### Appendix

#### Public Documentation License Notice

The contents of this Documentation are subject to the Public Documentation License Version 1.0 (the "License"); you may only use this Documentation if you comply with the terms of this License. A copy of the License is available at <http://www.openoffice.org/licenses/PDL.html>.

The Original Documentation is Calc – How To – FR – Formules et Fonctions.odt. The Initial Writer of the Original Documentation is Filip Lannoye Copyright CIRB-CIBG© 2006. All Rights Reserved. (Initial Writer contact(s): [flannoye@cirb.irisnet.be](mailto:flannoye@cirb.irisnet.be)).

Contributor(s): Roland Van Wunsel.  
Portions created by \_\_\_\_\_ are Copyright © \_\_\_\_\_ [Insert year(s)]. All Rights Reserved.  
(Contributor contact(s): [rvanwunsel@cibg.irisnet.be](mailto:rvanwunsel@cibg.irisnet.be)).

NOTE: The text of this Appendix may differ slightly from the text of the notices in the files of the Original Documentation. You should use the text of this Appendix rather than the text found in the Original Documentation for Your Modifications.

[Retour](#)